

COQ CHEATSHEET

Jules Jacobs

December 10, 2021

CONTENTS

1	Introduction	1
2	Goal tactics	2
3	Hypothesis tactics	3
4	Equality, rewriting, and computation rules	3
5	Inductive types and relations	3
5.1	Inductive types <code>Foo</code>	3
5.2	Inductive relations <code>Foo x y</code>	4
5.3	Getting the right induction hypothesis	4
6	Intro patterns	4
7	Forward reasoning	5
8	Composing tactics	5
9	Automation with <code>eauto</code>	5
10	Searching for lemmas and definitions	5
11	Common error messages	5

1 INTRODUCTION

This is Coq code that proves the strong induction principle for natural numbers:

```
From Coq Require Import Lia.

Lemma strong_induction (P : nat -> Prop) :
  (forall n, (forall m, m < n -> P m) -> P n) -> forall n, P n.
Proof.
  intros H n. eapply H. induction n.
  - lia.
  - intros m Hm. eapply H.
    intros k Hk. eapply IHn. lia.
Qed.
```

Coq proofs manipulate the *proof state* by executing a sequence of *tactics* such as `intros`, `eapply`, `induction`. Coq calculates the proof state for you after executing each tactic. Here's what Coq displays after executing the second `intros m Hm`:

```
P: nat -> Prop
H: forall n : nat, (forall m : nat, m < n -> P m) -> P n
n: nat
IHn: forall m : nat, m < n -> P m
m: nat
Hm: m < S n
-----
P m
```

The proof state consists of a list of variables and hypotheses above the line, and a goal below the line. A tactic may create 0, 1, 2, or more subgoals. A goal is solved if we successfully apply a tactic that creates no subgoals (such as the `lia` tactic). Some tactics create multiple subgoals, such as the `induction` tactic: it creates one subgoal for the base case of the induction, and one subgoal for the inductive case.¹ We have to solve all the subgoals with a bulleted list of tactic scripts:

```
tac1.
+ tac2.
+ tac3.
+ tac4.
```

Bullets can be nested by using different bullets for different levels (`-`, `+`, `*`):

```
tac1.
+ tac2.
  * tac3
  * tac4.
+ tac5.
```

We can also enter subgoals using brackets:

```
tac1.
{ tac2. }
{ tac3. }
tac4.
{ tac5. }
tac6.
```

This is most useful for solving side conditions. With bullets, we get a deep level of nesting if we have a sequence of tactics with side conditions. With brackets, we do not need to enclose the last subgoal in brackets, thus preventing deep nesting.

2 GOAL TACTICS

Goal	Tactic
$P \rightarrow Q$	<code>intros H</code>
$\neg P$	<code>intros H</code> (Coq defines $\neg P$ as $P \rightarrow \text{False}$)
$\forall x, P(x)$	<code>intros x</code>
$\exists x, P(x)$	<code>exists x, eexists</code>
$P \wedge Q$	<code>split</code>
$P \vee Q$	<code>left, right</code>
Q	<code>apply H, eapply H</code> (where $H : (...) \rightarrow Q$ is a lemma or hypothesis with conclusion Q)
False	<code>apply H, eapply H</code> (where $H : (...) \rightarrow \neg P$ is a lemma or hypothesis with conclusion $\neg P$)
Any goal	<code>exfalse</code> (turns any goal into False)
Skip goal	<code>admit</code> (skips goal so that you can work on other subgoals)

¹ Coq allows us to do induction not only on natural numbers, but also on other data types. Induction on other data types may create any number of subgoals, one for each constructor of the data type.

3 HYPOTHESIS TACTICS

Hypothesis	Tactic
$H : \text{False}$	<code>destruct H</code>
$H : \exists x, P(x)$	<code>destruct H as [x H]</code>
$H : P \wedge Q$	<code>destruct H as [H1 H2]</code>
$H : P \vee Q$	<code>destruct H as [H1 H2]</code>
$H : \forall x, P(x)$	<code>specialize (H y)</code>
$H : P \rightarrow Q$	<code>specialize (H G)</code> (where $G : P$ is a lemma or hypothesis)
$H : P$	<code>apply G in H, eapply G in H</code> (where $G : P \rightarrow (\dots)$ is a lemma or hypothesis)
$H : P, x : A$	<code>clear H, clear x</code> (remove hypothesis H or variable x)

4 EQUALITY, REWRITING, AND COMPUTATION RULES

Tactic	Meaning
<code>reflexivity</code>	Solve goal of the form $x = x$ or $P \leftrightarrow P$
<code>symmetry</code>	Turn goal $x = y$ into $y = x$ (or $P \leftrightarrow Q$)
<code>symmetry in H</code>	Turn hypothesis $H : x = y$ into $H : y = x$ (or $P \leftrightarrow Q$)
<code>unfold f</code>	Replace constant f with its definition (only in the goal)
<code>unfold f in H</code>	Replace constant f with its definition (in hypothesis H)
<code>unfold f in *</code>	Replace constant f with its definition (everywhere)
<code>simpl</code>	Rewrite with computation rules (in the goal)
<code>simpl in H</code>	Rewrite with computation rules (in hypothesis H)
<code>simpl in *</code>	Rewrite with computation rules (everywhere)
<code>rewrite H.</code>	Rewrite $H : x = y$ (in the goal).
<code>rewrite H in G.</code>	Rewrite $H : x = y$ (in hypothesis G).
<code>rewrite H in *.</code>	Rewrite H (everywhere).
<code>rewrite <-H.</code>	Rewrite $H : x = y$ backwards.
<code>rewrite H,G.</code>	Rewrite using H and then G .
<code>rewrite !H.</code>	Repeatedly rewrite using H .
<code>rewrite ?H.</code>	Try rewriting using H .
<code>subst</code>	Substitute away all equations $H : x = A$ with a variable on one side.
<code>injection H as H</code>	Use injectivity of C to turn $H : C\ x = C\ y$ into $H : x = y$.
<code>discriminate H</code>	Solve goal with inconsistent assumption $H : C\ x = D\ y$.
<code>simplify_eq</code>	Automated tactic that does <code>subst</code> , <code>injection</code> , and <code>discriminate</code> automatically.

5 INDUCTIVE TYPES AND RELATIONS

5.1 Inductive types Foo

Hypothesis / Term	Tactic
$H : \text{Foo}$	<code>destruct H as [a b c d e f]</code>
$H : \text{Foo}$	<code>induction H as [a b IH c d e IH1 IH2 f IH]</code>

5.2 Inductive relations $\text{Foo } x \ y$

Goal	Tactic
$\text{Foo } x \ y$	<code>constructor</code> , <code>econstructor</code> (tries to solve goal by applying all constructors of <code>Foo</code>)
Hypothesis	Tactic
$H : \text{Foo } x \ y$	<code>inversion H</code> (use when x, y are fixed terms)
$H : \text{Foo } x \ y$	<code>induction H</code> (use when x, y are variables)

It is often useful to define the tactic `Ltac inv H := inversion H; clear H; subst.` and use this instead of `inversion`.

5.3 Getting the right induction hypothesis

The `revert` tactic is useful to obtain the correct induction hypothesis:

Hypothesis	Tactic
$H : P$	<code>revert H</code> (opposite of <code>intros H</code> : turn goal Q into $P \rightarrow Q$)
$x : A$	<code>revert x</code> (opposite of <code>intros x</code> : turn goal Q into $\forall x, Q$)

A common pattern is `revert x. induction n; intros x; simpl.` A good rule of thumb is that you should create a separate lemma for each inductive argument, so that `induction` is only ever used at the start of a lemma (possibly preceded by some `revert`).

6 INTRO PATTERNS

The `destruct x as pat` and `intros pat` tactics can unpack multiple levels at once using nested *intro patterns*. The `intros` tactic can also be chained: `intros x y z. \equiv intros x. intros y. intros z.`

Data	Pattern
$\exists x, P$	<code>[x H]</code>
$P \wedge Q$	<code>[H1 H2]</code>
$P \vee Q$	<code>[H1 H2]</code>
False	<code>[]</code>
$A * B$	<code>[x y]</code>
$A + B$	<code>[x y]</code>
option A	<code>[x]</code>
nat	<code>[n]</code>
list A	<code>[x xs]</code>
Inductive type	<code>[a b c d e f]</code>
Inductive type	<code>[]</code> (unpack with names chosen by Coq)
$x = y$	<code>-></code> or <code><-</code> (substitute the equality)
Any	<code>?</code> (introduce variable/hypothesis with name chosen by Coq)

Furthermore, $(x \ \& \ y \ \& \ z \ \& \ \dots)$ is equivalent to `[x [y [z ...]]]`.

Because $\exists x, P$, $P \wedge Q$, $P \vee Q$, False are *defined* as inductive types, their intro patterns are special cases of the intro pattern for inductive types, and you can also use the `[]` intro pattern for them.

7 FORWARD REASONING

Tactic	Meaning
<code>assert P as H</code>	Create new hypothesis $H : P$ after proving subgoal P
<code>cut P</code>	Split goal Q into two subgoals $P \rightarrow Q$ and P

Intro patterns can be used in combination with the `assert` tactic, e.g. `assert (A = B) as ->` or `assert (exists x, P) as [x H]`.

8 COMPOSING TACTICS

Tactic	Meaning
<code>tac1; tac2</code>	Do <code>tac2</code> on all subgoals created by <code>tac1</code> .
<code>tac1; [tac2 ..]</code>	Do <code>tac2</code> only on the first subgoal.
<code>tac1; [.. tac2]</code>	Do <code>tac2</code> only on the last subgoal.
<code>tac1; [tac2 .. tac3 tac4]</code>	Do tactics on corresponding subgoals.
<code>tac1; [tac2 tac3.. tac4]</code>	Do tactics on corresponding subgoals.
<code>tac1 tac2</code>	Try <code>tac1</code> and if it fails do <code>tac2</code> .
<code>try tac1</code>	Try <code>tac1</code> , and do nothing if it fails.
<code>repeat tac1</code>	Repeatedly do <code>tac1</code> until it fails.
<code>progress tac1</code>	Do <code>tac1</code> and fail if it does nothing.

9 AUTOMATION WITH `eauto`

The `eauto` tactic tries to solve goals using `eapply`, `reflexivity`, `eexists`, `split`, `left`, `right`. You can specify the search depth using `eauto n` (the default is $n = 5$).

You can give `eauto` additional lemmas to use with `eauto using lemma1, lemma2`. You can also use `eauto using foo` where `foo` is an inductive type. This will use all the constructors of `foo` as lemmas.

10 SEARCHING FOR LEMMAS AND DEFINITIONS

TODO

11 COMMON ERROR MESSAGES

TODO

Please submit your errors to me so that I can add them to this section.

You can also suggest additional content.

For instance:

- Installing Coq
- Compilation and multiple files
- Definition, Fixpoint, Inductive
- Implicit arguments
- Searching for lemmas

- Hint databases
- `match_goal`
- Type classes
- `setoid_rewrite`
- `CoInductive`, `cofix` (and `fix`)
- Mutually inductive lemmas
- `ssreflect`
- `stdpp`
- Modules

`julesjacobs@gmail.com`